Newsgroup: it.comp.lang.visual-basic
Titolo: Rappresentazione di tabelle...

Ciao ragazzi, ho un problema non indifferente!
Devo rappresentare una tabella in una form, avevo pensato di farlo tramite
foglio di excel che ero riuscito ad importare con gli Office Web Components,
Purtroppo però questi componenti non supportano la stampa.
Qualcuno ha qualche suggerimento da darmi per rappresentare una tabella?
(nb: Devo riuscire a colorare le celle ecc...)
Oppure qualcuno sa come potrei stampare comunque l'oggetto?
Grazie a chiunque mi aiuti.. sono nei casini!
--------------------------------
Inviato via 
http://arianna.libero.it/usenet/

 Oppure qualcuno sa come potrei stampare comunque l'oggetto?
cosě, come consiglio, direi di separare la fase di visualizzazine da quella
della stampa.
se hai bisogno di avere un controllo molto "fine" sull'aspetto della griglia
in visualizzazione, consiglierei la buona vecchia flex grid. devi scrivere
un po' di codice, ma alla fine ci fai quello che vuoi.
per la stampa... printer.print :-) anche qui, tanto da scrivere ma totale
controllo sull'output.
-- 
Salutoni
Sergio
per rispondere... togli il group dall'indirizzo

Ma visto che al momento la visualizzazione ce l'ho già, per stampare il
controllo riuscirei con un print della form?
Se si come faccio ad adattare alla pagina la stampa della form?
Ho trovato un sorgente sul sito comune ma purtroppo non mi funziona!
Grazie mille!


 Ho trovato un sorgente sul sito comune ma purtroppo non mi funziona!
http://makeashorterlink.com/?N4E121287

 
http://makeashorterlink.com/?N4E121287
Ciao Lupo, grazie per la dritta, ma provandola ho ancora un problema...
Alla riga: Pic1 = CaptureForm(Screen.ActiveForm)
mi dice "Utilizzo non valido di Property"
Io ho definito Pic1 come Picture, anche se non mi viene suggerito, ma
battendo invio lo accetta...
devo aggiungere qualche riferimento??
Help!


 devo aggiungere qualche riferimento??
Se hai letto tutto il 3d (mancava un po' di roba), deve andare.
Io ho appena riprovato, sincerandomi che Pic1 sia una PictureBox.
Sicuro inoltre di avere una Screen.ActiveForm?

 Screen.ActiveForm?
Non so di che parlate ma a leggervi mi sa che parlate di due cose diverse.
Skini82 ha definito Pic1 come Picture, mentre Lupo parla di PictureBox
Ocio =)
Lokar
-- 
Sito Comune: 
http://www.it-lang-vb.net/
Manifesto: 
http://www.it-lang-vb.net/docs/Manifesto.txt
Statuto: 
http://www.it-lang-vb.net/docs/Statuto.txt
UGIdotNET - 
http://www.ugidotnet.org

Mi da SUB o FUNCTION non definita...
Mi manca la funzione GetDC e anche GetWindowDC...
Può essere??
Ciao e ancora grazie x l'aiuto!


 PictureBox
Po'esse!
Mai definito una Picture in viata mia.
Mi rendo conto che scrivere "Inizialmente faccio (appoggiandomi a una
Picture qualsiasi):" possa essere frainteso, ma intentendevo riferirmi a un
controllo picturebox definito su un form, anche se invisibile e di qualsiasi
dimensione. Inoltre, quel topic è fatto di 3 post e il terzo cita la roba
che mancava nel primo.

Dimenticavo...
La definizione l'ho sistemata! era un problemuccio stupido!
Cmq adesso ho quell'altro problema! ciao e grazie a tutti quantiii!


 Puň essere??
Sě, abbi pazienza, ma quella cosa era stata estrapolata dal contesto e
evidentemente non č stato un lavoro ben fatto. Lo rifaccio. L'ho provato e
va:
'NEL FORM:
'#########
Private Sub btn_stampa_finestra_Click()
'Inizialmente faccio (appoggiandomi a una PictureBox, pic1, qualsiasi):
    Set pic1 = CaptureForm(Screen.ActiveForm)
'Poi:
    amplia = vbYes
    Call PrintPictureToFitPage(Printer, pic1, IIf(amplia = vbYes, 0, 2 ^ 0))
    Printer.EndDoc
End Sub
'   Informazioni per gestione Catture
'---------------------------------------------------------------------------
Public Function CreateBitmapPicture(ByVal hBmp As Long, ByVal hPal As Long)
As Picture
    ncar = "CreateBitmapPicture"
' CreateBitmapPicture
'    - Creates a bitmap type Picture object from a bitmap and
'      palette.
'
' hBmp
'    - Handle to a bitmap.
'
' hPal
'    - Handle to a Palette.
'    - Can be null if the bitmap doesn't use a palette.
'
' Returns
'    - Returns a Picture object containing the bitmap.
    Dim r As Long
    Dim pic As PicBmp
'       IPicture requires a reference to "Standard OLE Types."
    Dim IPic As IPicture
    Dim IID_IDispatch As GUID
'       Fill in with IDispatch Interface ID.
    With IID_IDispatch
        .Data1 = &H20400
        .Data4(0) = &HC0
        .Data4(7) = &H46
    End With
'       Fill Pic with necessary parts.
    With pic
        .Size = Len(pic)          ' Length of structure.
        .Type = vbPicTypeBitmap   ' Type of Picture (bitmap).
        .hBmp = hBmp              ' Handle to bitmap.
        .hPal = hPal              ' Handle to palette (may be null).
    End With
'       Create Picture object.
    r = OleCreatePictureIndirect(pic, IID_IDispatch, 1, IPic)
'       Return the new Picture object.
    Set CreateBitmapPicture = IPic
End Function
Public Function CaptureWindow(ByVal hWndSrc As Long, ByVal Client As
Boolean, ByVal LeftSrc As Long, ByVal TopSrc As Long, ByVal WidthSrc As
Long, ByVal HeightSrc As Long) As Picture
    ncar = "CaptureWindow"
' CaptureWindow
'    - Captures any portion of a window.
'
' hWndSrc
'    - Handle to the window to be captured.
'
' Client
'    - If True CaptureWindow captures from the client area of the
'      window.
'    - If False CaptureWindow captures from the entire window.
'
' LeftSrc, TopSrc, WidthSrc, HeightSrc
'    - Specify the portion of the window to capture.
'    - Dimensions need to be specified in pixels.
'
' Returns
'    - Returns a Picture object containing a bitmap of the specified
'      portion of the window that was captured.
    Dim hDCMemory As Long
    Dim hBmp As Long
    Dim hBmpPrev As Long
    Dim r As Long
    Dim hDCSrc As Long
    Dim hPal As Long
    Dim hPalPrev As Long
    Dim RasterCapsScrn As Long
    Dim HasPaletteScrn As Long
    Dim PaletteSizeScrn As Long
    Dim LogPal As LOGPALETTE
'       Depending on the value of Client get the proper device context.
    If Client Then
        hDCSrc = GetDC(hWndSrc) ' Get device context for client area.
    Else
        hDCSrc = GetWindowDC(hWndSrc) ' Get device context for entire
                                      ' window.
    End If
'       Create a memory device context for the copy process.
    hDCMemory = CreateCompatibleDC(hDCSrc)
'       Create a bitmap and place it in the memory DC.
    hBmp = CreateCompatibleBitmap(hDCSrc, WidthSrc, HeightSrc)
    hBmpPrev = SelectObject(hDCMemory, hBmp)
'       Get screen properties.
    RasterCapsScrn = GetDeviceCaps(hDCSrc, RASTERCAPS)      ' Raster
                                                            ' capabilities.
    HasPaletteScrn = RasterCapsScrn And RC_PALETTE          ' Palette
                                                            ' support.
    PaletteSizeScrn = GetDeviceCaps(hDCSrc, SIZEPALETTE)    ' Size of
                                                            ' palette.
'       If the screen has a palette make a copy and realize it.
    If HasPaletteScrn And (PaletteSizeScrn = 256) Then
'           Create a copy of the system palette.
        LogPal.palVersion = &H300
        LogPal.palNumEntries = 256
        r = GetSystemPaletteEntries(hDCSrc, 0, 256, LogPal.palPalEntry(0))
        hPal = CreatePalette(LogPal)
'           Select the new palette into the memory DC and realize it.
        hPalPrev = SelectPalette(hDCMemory, hPal, 0)
        r = RealizePalette(hDCMemory)
    End If
'       Copy the on-screen image into the memory DC.
    r = BitBlt(hDCMemory, 0, 0, WidthSrc, HeightSrc, hDCSrc, LeftSrc,
TopSrc, vbSrcCopy)
'       Remove the new copy of the  on-screen image.
    hBmp = SelectObject(hDCMemory, hBmpPrev)
'       If the screen has a palette get back the palette that was
'       selected in previously.
    If HasPaletteScrn And (PaletteSizeScrn = 256) Then
        hPal = SelectPalette(hDCMemory, hPalPrev, 0)
    End If
'       Release the device context resources back to the system.
    r = DeleteDC(hDCMemory)
    r = ReleaseDC(hWndSrc, hDCSrc)
'       Call CreateBitmapPicture to create a picture object from the
'       bitmap and palette handles. Then return the resulting picture
'       object.
    Set CaptureWindow = CreateBitmapPicture(hBmp, hPal)
End Function
Public Function CaptureScreen() As Picture
    ncar = "CaptureScreen"
' CaptureScreen
'    - Captures the entire screen.
'
' Returns
'    - Returns a Picture object containing a bitmap of the screen.
    Dim hWndScreen As Long
'       Get a handle to the desktop window.
    hWndScreen = GetDesktopWindow()
'       Call CaptureWindow to capture the entire desktop give the handle
'       and return the resulting Picture object.
    Set CaptureScreen = CaptureWindow(hWndScreen, False, 0, 0, Screen.Width
\ Screen.TwipsPerPixelX, Screen.Height \ Screen.TwipsPerPixelY)
End Function
Public Function CaptureForm(frmSrc As Form) As Picture
    ncar = "CaptureForm"
' CaptureForm
'    - Captures an entire form including title bar and border.
'
' frmSrc
'    - The Form object to capture.
'
' Returns
'    - Returns a Picture object containing a bitmap of the entire
'      form.
'       Call CaptureWindow to capture the entire form given its window
'       handle and then return the resulting Picture object.
    Set CaptureForm = CaptureWindow(frmSrc.hwnd, False, 0, 0,
frmSrc.ScaleX(frmSrc.Width, vbTwips, vbPixels), frmSrc.ScaleY(frmSrc.Height,
vbTwips, vbPixels))
End Function
Public Function CaptureClient(frmSrc As Form) As Picture
    ncar = "CaptureClient"
' CaptureClient
'    - Captures the client area of a form.
'
' frmSrc
'    - The Form object to capture.
'
' Returns
'    - Returns a Picture object containing a bitmap of the form's
'      client area.
'       Call CaptureWindow to capture the client area of the form given
'       its window handle and return the resulting Picture object.
    Set CaptureClient = CaptureWindow(frmSrc.hwnd, True, 0, 0,
frmSrc.ScaleX(frmSrc.ScaleWidth, frmSrc.ScaleMode, vbPixels),
frmSrc.ScaleY(frmSrc.ScaleHeight, frmSrc.ScaleMode, vbPixels))
End Function
Public Function CaptureActiveWindow() As Picture
    ncar = "CaptureActiveWindow"
' CaptureActiveWindow
'    - Captures the currently active window on the screen.
'
' Returns
'    - Returns a Picture object containing a bitmap of the active
'      window.
    Dim hWndActive As Long
    Dim r As Long
    Dim RectActive As RECT
'       Get a handle to the active/foreground window.
    hWndActive = GetForegroundWindow()
'       Get the dimensions of the window.
    r = GetWindowRect(hWndActive, RectActive)
'       Call CaptureWindow to capture the active window given its
'       handle and return the Resulting Picture object.
    Set CaptureActiveWindow = CaptureWindow(hWndActive, False, 0, 0,
RectActive.Right - RectActive.Left, RectActive.Bottom - RectActive.Top)
End Function
Public Sub PrintPictureToFitPage(Prn As Object, pic As Picture, Optional
parametri = 0)
'Public Sub PrintPictureToFitPage(Prn As Printer, Pic As Picture, Optional
parametri = 0)
    ncar = "PrintPictureToFitPage"
' PrintPictureToFitPage
'    - Prints a Picture object as big as possible.
'
' Prn
'    - Destination Printer object.
'
' Pic
'    - Source Picture object.
' parametri   2^0   Stampa senza ampliare
    Const vbHiMetric As Integer = 8
    Dim PicRatio As Double
    Dim PrnWidth As Double
    Dim PrnHeight As Double
    Dim PrnRatio As Double
    Dim PrnPicWidth As Double
    Dim PrnPicHeight As Double
    If (parametri And 2 ^ 0) Then
        Prn.PaintPicture pic, 0, 0
        GoTo esce
    End If
'       Determine if picture should be printed in landscape or portrait
'       and set the orientation.
    'If Prn Is Printer Then
    '    If Pic.Height >= Pic.Width Then
    '        Prn.Orientation = vbPRORPortrait   ' Taller than wide.
    '    Else
    '        Prn.Orientation = vbPRORLandscape  ' Wider than tall.
    '    End If
    'End If
'       Calculate device independent Width-to-Height ratio for picture.
    PicRatio = pic.Width / pic.Height
'       Calculate the dimentions of the printable area in HiMetric.
    PrnWidth = Prn.ScaleX(Prn.ScaleWidth, Prn.ScaleMode, vbHiMetric)
    PrnHeight = Prn.ScaleY(Prn.ScaleHeight, Prn.ScaleMode, vbHiMetric)
'       Calculate device independent Width to Height ratio for printer.
    PrnRatio = PrnWidth / PrnHeight
'       Scale the output to the printable area.
    If PicRatio >= PrnRatio Then
'           Scale picture to fit full width of printable area.
        PrnPicWidth = Prn.ScaleX(PrnWidth, vbHiMetric, Prn.ScaleMode)
        PrnPicHeight = Prn.ScaleY(PrnWidth / PicRatio, vbHiMetric,
Prn.ScaleMode)
    Else
'           Scale picture to fit full height of printable area.
        PrnPicHeight = Prn.ScaleY(PrnHeight, vbHiMetric, Prn.ScaleMode)
        PrnPicWidth = Prn.ScaleX(PrnHeight * PicRatio, vbHiMetric,
Prn.ScaleMode)
    End If
'       Print the picture using the PaintPicture method.
    Prn.PaintPicture pic, 0, 0, PrnPicWidth, PrnPicHeight
esce:
End Sub
'IN UN MODULO (BAS):
'###################
Public Const RASTERCAPS As Long = 38
Public Const RC_PALETTE As Long = &H100
Public Const SIZEPALETTE As Long = 104
Declare Function OleCreatePictureIndirect Lib "olepro32.dll" (PicDesc As
PicBmp, RefIID As GUID, ByVal fPictureOwnsHandle As Long, IPic As IPicture)
As Long
Declare Function GetDC Lib "user32" (ByVal hwnd As Long) As Long
Declare Function GetWindowDC Lib "user32" (ByVal hwnd As Long) As Long
Declare Function CreateCompatibleBitmap Lib "gdi32" (ByVal hDC As Long,
ByVal nWidth As Long, ByVal nHeight As Long) As Long
Declare Function CreateCompatibleDC Lib "gdi32" (ByVal hDC As Long) As Long
Declare Function SelectObject Lib "gdi32" (ByVal hDC As Long, ByVal hObject
As Long) As Long
Declare Function SelectPalette Lib "gdi32" (ByVal hDC As Long, ByVal
hPalette As Long, ByVal bForceBackground As Long) As Long
Declare Function CreatePalette Lib "gdi32" (lpLogPalette As LOGPALETTE) As
Long
Declare Function GetDesktopWindow Lib "user32" () As Long
Declare Function GetDeviceCaps Lib "gdi32" (ByVal hDC As Long, ByVal
iCapabilitiy As Long) As Long
Declare Function GetForegroundWindow Lib "user32" () As Long
Declare Function GetSystemPaletteEntries Lib "gdi32" (ByVal hDC As Long,
ByVal wStartIndex As Long, ByVal wNumEntries As Long, lpPaletteEntries As
PALETTEENTRY) As Long
Declare Function GetWindowRect Lib "user32" (ByVal hwnd As Long, lpRect As
RECT) As Long
Declare Function RealizePalette Lib "gdi32" (ByVal hDC As Long) As Long
Declare Function ReleaseDC Lib "user32" (ByVal hwnd As Long, ByVal hDC As
Long) As Long
Declare Function BitBlt Lib "gdi32" (ByVal hDCDest As Long, ByVal XDest As
Long, ByVal YDest As Long, ByVal nWidth As Long, ByVal nHeight As Long,
ByVal hDCSrc As Long, ByVal XSrc As Long, ByVal YSrc As Long, ByVal dwRop As
Long) As Long
Declare Function DeleteDC Lib "gdi32" (ByVal hDC As Long) As Long
Type PALETTEENTRY
    peRed As Byte
    peGreen As Byte
    peBlue As Byte
    peFlags As Byte
End Type
Type LOGPALETTE
    palVersion As Integer
    palNumEntries As Integer
    palPalEntry(255) As PALETTEENTRY  ' Enough for 256 colors.
End Type
Type GUID
    Data1 As Long
    Data2 As Integer
    Data3 As Integer
    Data4(7) As Byte
End Type
Type RECT
    Left As Long
    Top As Long
    Right As Long
    Bottom As Long
End Type
Type PicBmp
    Size As Long
    Type As Long
    hBmp As Long
    hPal As Long
    Reserved As Long
End Type

 Cmq adesso ho quell'altro problema! ciao e grazie a tutti quantiii!
Vedi il post sopra. C'è tutto il codice completo e provato.

Fantastico! Grazie!


Fantastico! Grazie!


 Fantastico! Grazie!
Ma ti funziona o no?